```
> with(combinat):
Warning, the protected name Chi has been redefined and unprotected
```

6 to the power of 10:

```
> 6^10;
```

$$60466176$$

factorial:

```
> 5!;
```

$$120$$

number of combinations (say 20 'choose' 5):

```
> binomial(20,5);
```

$$15504$$

number of permutations (20 P 5) can be indirectly computed by:

```
> binomial(20,5)*5!;
```

$$1860480$$

the multinomial coefficient (say 20 'choose' 5,4,4,3,2,2):

```
> multinomial(20,5,4,4,3,2,2);
```

$$1466593128000$$

DISCRETE DISTRIBUTIONS

```
> values:=[-4,0,5,20]: pr:=[0.43,0.31,0.23,0.03]:
```

Mean, variance and standard deviation:

```
> m:=add(values[i]*pr[i],i=1..4); add(values[i]^2*pr[i],i=1..4)-m^2;
  sigma:=sqrt(%);
```

$$m := 0.03$$

$$24.6291$$

$$\sigma := 4.962771403$$

skewness and kurtosis

```
> add((values[i]-m)^3*pr[i],i=1..4)/sigma^3;
```

$$1.955461875$$

```
> add((values[i]-m)^4*pr[i],i=1..4)/sigma^4;
```

$$8.284004246$$

PGF:

```
> PGF:=add(z^values[i]*pr[i],i=1..4);
```

$$PGF := 0.31 + \frac{0.43}{z^4} + 0.23\, z^5 + 0.03\, z^{20}$$

```
> m:=subs(z=1,diff(PGF,z));    # getting mean and variance based on
  PGF
  subs(z=1,diff(PGF,z,z))+m-m^2;
```

$$m := 0.03$$

$$24.6291$$

Probability that independent sum of 100 of these will have a value less than 50:

```
> aux:=expand(PGF^100):
  add(coeff(aux,z,i),i=-400..49);
```

0.8268463361

plotting the corresponding distribution:

```
> with(plots):     pointplot([seq([i,coeff(aux,z,i)],i=-140..150)]);
Warning, the name changecoords has been redefined
```



COMPUTING PROBABILITIES RELATED TO 'SPECIAL' DISCRETE DISTRIBUTIONS:

BINOMIAL (getting more than 3 sixes in 20 rolls of a die):

```
> add(binomial(20,i)*(1/6.)^i*(5/6)^(20-i),i=4..20);
```

0.4334543627

NEGATIVE BINOMIAL (need more than 30 rolls to get the fourth six):
Either directly:

```
> 1-add(binomial(i-1,4-1)*(1/6.)^4*(5/6)^(i-4),i=4..30);
```

0.2396195261

Or through the binomial formula (getting fewer than 4 sixes in 30 rolls):

```
> add(binomial(30,i)*(1/6.)^i*(5/6)^(30-i),i=0..3);
```

0.2396195270

Or by expanding the corresponding PGF:

```
> PGF:=(z/6./(1-5/6*z))^4:     # use at least one decimal point !!!
  aux:=series(PGF,z,75):    # when expanding non-polynomial PDF, we
  have to use 'series'
  1-add(coeff(aux,z,i),i=4..30);
```

0.2396195261

and plotting the distribution

```
> with(plots): pointplot([seq([i,coeff(aux,z,i)],i=4..74)]);
```



HYPERGEOMETRIC (getting at least 4 spades when dealt 10 cards):

```
> add(binomial(13,i)*binomial(39,10-i)/binomial(52,10.),i=4..10);
```

0.2042716078

POISSON (getting at least 8 arrivals during the next 15 minutes, when the arrival rate is 27 per hour):

```
> L:=27*15/60.:    1-add(L^i/i!*exp(-L),i=0..7);
```

0.3640918365

MULTINOMIAL (getting exactly 3 sixes and 2 fives in 20 rolls):

```
> multinomial(20,3,2,15)*(1/6.)^3*(1/6)^2*(4/6)^15;
```

0.04553219867

PGF of W=2X-3Y+4, where X is the number of sixes and Y is the number of ones

```
> P:=add(add(multinomial(20,i,j,20-i-j)*(1/6.)^i*(1/6)^j*(4/6)^(20-i
  -j)*z^(2*i-3*j+4),j=0..20-i),i=0..20):
```

Probability that W>5:

```
> add(coeff(P,z,i),i=6..20);
```

0.2345294128

MULTIVARIATE HYPERGEOMETRIC (getting exactly 4 spades and 3 diamonds when dealt 10 cards):

```
> binomial(13,4)*binomial(13,3)*binomial(26,3)/binomial(52,10.);
```

0.03360766030

PGF of W=2X-3Y+4, where X is the number of spades and Y the number of diamonds:

```
> P:=add(add(binomial(13,i)*binomial(13,j)*binomial(26,10-i-j)/binom
  ial(52,10.)*z^(2*i-3*j+4),j=0..10-i),i=0..10):
```

Probability of W>5:

```
> add(coeff(P,z,i),i=6..24);
```

0.2255859149

UNIVARIATE CONTINUOUS DISTRIBUTIONS

Given f(x)

```
> f:=piecewise(x<-1,0,  x<1,c*(x+1),  x<3,c*(3-x),  0);
```

$$f := \begin{cases} 0 & x < -1 \\ c\,(x+1) & x < 1 \\ c\,(3-x) & x < 3 \\ 0 & otherwise \end{cases}$$

find c:

```
> int(f,x=-1..3);
```

$$4\,c$$

```
> c:=1/4:
> plot(f,x=-2..4);
```



find F(y):

```
> F:=int(f,x=-infinity..y);
```

$$F := \begin{cases} 0 & y \le -1 \\ \frac{1}{8}y^2 + \frac{1}{4}y + \frac{1}{8} & y \le 1 \\ \frac{3}{4}y - \frac{1}{8}y^2 - \frac{1}{8} & y \le 3 \\ 1 & 3 < y \end{cases}$$

```
> plot(F,y=-2..4);
```



mean:

```
> m:=int(x*f,x=-1..3);
```

$$m := 1$$

variance:

```
> int(x^2*f,x=-1..3)-m^2;
```

$$\frac{2}{3}$$

MGF:

```
> M:=simplify(int(f*exp(x*t),x=-1..3));
```

$$M := \frac{1}{4}\,\frac{\left(\mathbf{e}^{(4\,t)} - 2\,\mathbf{e}^{(2\,t)} + 1\right)\mathbf{e}^{(-t)}}{t^2}$$

median:

```
> fsolve(F=1/2);
```

$$1.000000000$$